# $Id: PKGBUILD 90055 2013-05-06 19:38:09Z foutrelis $
# Maintainer: Federico Cinelli <cinelli@aur.archlinux.org>
# Maintainer: Mateusz Herych   <heniekk@gmail.com>

pkgname=libquicktime
pkgver=1.2.4
pkgrel=5
pkgdesc="A library for reading and writing quicktime files"
arch=('i686' 'x86_64')
license=('GPL')
url="http://libquicktime.sourceforge.net/"
depends=('gtk2' 'ffmpeg' 'libxv' 'libxaw' 'faad2' 'faac')
makedepends=('mesa-libgl')
options=('!libtool')
source=("http://downloads.sourceforge.net/sourceforge/$pkgname/$pkgname-$pkgver.tar.gz")
sha256sums=('1c53359c33b31347b4d7b00d3611463fe5e942cae3ec0fefe0d2fd413fd47368')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  ./configure --prefix=/usr \
  	--enable-gpl \
	--with-ffmpeg \
	--with-x264 \
	--without-doxygen
	make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
